
Kritériumok használata Excelben 
Azokban az Excel függvényekben, ahol kritérium megadására van szükség, több lehetőségünk is 

adódik. Kritérium lehet konkrét szöveg vagy szám, logikai feltétel, vagy akár bonyolultabb 

cellatartalom is. 

Példák, mi írható a kritérium1…n helyére és hogyan: 

Ha azt akarjuk megszámolni, hogy 

-hány alapanyagunk tárolandó KLT-ben a kritérium: ’’KLT” 

-hány KLT-ben tudunk 200-nál több terméket tárolni a kritérium: ’’>=200” de ha már a 200 egy 

külön cellában van pl. B1-ben akkor : ’’>’’&B1 

-hány db művelethez tartozó műveleti idő nagyobb, mint az átlag: ’’>”&ÁTLAG(B:B) 

-hány db művelet tartozik a C gyártósorhoz? ’’C*” 

A szöveges értékeket minden esetben felső idézőjelek (’’ ’’) közé szükséges tenni, de ha már 

cellahivatkozást is tartalmaz a képlet, akkor használni kell a & karaktert az összefűzéshez. 

A * karakterhelyettesítő segítségével bármennyi karaktert jelölhetünk, de fontos, hogy hol 

helyezkedik el. (Ha ’’C*’’-t adunk meg, akkor a C-t követheti bármennyi és bármilyen karakter, de 

a C-t nem előzheti meg semmi. Ha ’’*C’’-t használunk, akkor a C előtt lehet bármennyi és bármilyen 

karakter de a C-t már nem követheti semmi…stb.) 

A ? karakterhelyettesítő segítségével pontosan annyi karakter helyettesíthető, ahány kérdőjelet 

használunk. (Ha ’’C?’’ -t használunk, akkor a C1…C9 műveleteket használja fel. ’’C??” esetén már 

C10…C99-ig számol. ’’?C?” esetén a QC1…QC9 műveletek kaphatók vissza a házi feladat 

esetében.) 

 

SZUMHATÖBB 

– több feltételnek megfelelő argumentum összeadása 

Szintaxis: SZUMHATÖBB(összegtartomány; kritériumtartomány1; kritérium1; 

[kritériumtartomány2; kritérium2]; ...) 

Akkor használhatjuk, ha egy vagy több kritériumnak megfelelő értéket szeretnénk összeadni. 

Először ki kell jelölni az összegtartományt, vagyis azt a tartományt (sor, oszlop, mátrix..), amelyen 

belül vannak az összeadandó értékek. Ezt követően pontosvesszővel elválasztva kiválasztjuk a 

kritériumtartományt (sor, oszlop, mátrix… az összegtartománynak megfelelően, de akár maga az 

összegtartomány is lehet a kritériumtartomány). Végül pedig magát a kritériumot adjuk meg. (Ha 

több kritériumnak megfelelő értékeket szeretnénk összegezni, akkor a kritérium1 után 

pontosvesszővel megadjuk a következő kritériumtartományt és kritériumot…és így tovább.) 

 

Példa: Határozzuk meg az egyes gyártósorokhoz tartozó műveleti idők összegét! 

Amennyiben a teljes üzemre vonatkozó összes műveleti időt vesszük kiindulásnak, akkor a 

SZUMHATÖBB függvénnyel megadhatjuk az egyes gyártósorok összes műveleti idejét, annak a 

vizsgálatával, hogy milyen betűvel jelölt gyártósorhoz tartoznak a műveletek. 

=SZUMHATÖBB($B$2:$B$39;$A$2:$A$39;"C??")+SZUMHATÖBB($B$2:$B$39;$A$2:$A$3

9;"C?") 

A függvényben először az összegtartományt kell kijelölni, ami a műveleti időket tartalmazó oszlop 

(B2:B39). 

Ezt követően a kritériumtartományt kell kijelölni: ez a műveletek megnevezését tartalmazó oszlop 

(A2:A39). 

A következő lépés a kritérium meghatározása. Az első oldalon leírtak alapján használjuk a ? 

karakterhelyettesítőt. 

Ebben az esetben összegeznünk kell a ’’C?” formátumú cellákat ( C1-C9), majd összegezzük a 

’’C??” formátumú cellákat is (C10-C13). Végül a két függvény eredményének összege fogja 

megadni nekünk a C gyártósorhoz tartozó összes műveleti idő értékét. 

 

 

 

 

DARABHATÖBB 
- több feltételnek megfelelő argumentum darabszámának megadása 

Szintaxis: DARABHATÖBB(kritériumtartomány1; kritérium1; [kritériumtartomány2; kritérium2]; 

…) 

A SZUMHATÖBB-höz hasonlóan itt is több kritérium megadható, aminek mindnek teljesülnie kell, 

hogy a függvény összeadja az érintett cellák számosságát. Tehát ez a függvény azt adja eredményül, 

hogy hány cellára teljesül(nek) a feltétel(ek). 

Példa: Számoljuk ki a maximális követőműveletszámot az egyes gyártósorokra! 

A gyártosorokhoz tartozó követő műveletek maximális értéke egyenlő a gyártósori műveletek 

darabszámánál eggyel kisebb számmal. 

=DARABHATÖBB($A$2:$A$39;"C*")-1 

A függvényben először a kritériumtartomány kell megadni. Esetünkben az üzem összes 

műveletének elnevezését tartalmazó oszlopot (A2:A39). 

Ezt követően meg kell fogalmazni a kritériumot, ami esetünkben a C-s műveleteket jelenti ’’C*”. A 

C* karakterhelyettesítő bármennyi karaktert lefedhet, így ebben a kritériumban benne vannak C1-

től C13-ig a műveletek. 

Végül a kapott darabszámot, ami megadja, hogy a kritériumtartományból hány darab kezdődik C 

betűvel, eggyel csökkenteni kell, hogy megkapjuk a maximális követő műveletek számát. 

 

 

 

TRANSZPONÁLÁS 
– több cella elforgatása (pl. oszlop(ok)ból sor(ok) készítése) 

Szintaxis: TRANSZPONÁLÁS(tömb) 

A TRANSZPONÁLÁS függvény segítségével sorokból oszlopokat és oszlopokból sorokat érhetünk 

el az Excelben. Ehhez először ki kell jelölni a cellatartományt, ahova be szeretnénk illeszteni 

elforgatva a meglévő celláinkat. A cellatartomány kijelölése után nem kattintunk el sehova, csak 

rögtön elkezdjük begépelni a ,, =TRANSZPONÁLÁS(’’ karaktereket. Ha ezzel megvagyunk, akkor 

kijelöljük az elforgatni kívánt cellatartományt és begépelünk utána egy bezáró zárójelet )” , majd 

lenyomjuk a CTRL+SHIFT+ENTER billentyűkombinációt. Ennek a segítségével megjelenik a {}- 

zárójel a beírt képletben és a teljes cellatartomány (tömb) elforgatása került. 

KEREK.FEL, KEREK.LE 

– Adott számjegyek számára történő felfelé vagy lefelé kerekítés 

Szintaxis: KEREK.LE(szám; hány_számjegy) vagy KEREK.FEL(szám; hány_számjegy) 

A képletben a szám helyére a kerekíteni kívánt szám celláját kell megadni, míg a pontosvessző után 

a hány_számjegy helyére azt a számot kell megadni, hogy hány tizedesjegyre történjen a kerekítés. 

(pl. KEREK.FEL(10,2;0)→11) 

HAHIBA 
– Hiba esetén visszaadandó érték meghatározása 

Szintaxis: HAHIBA(érték; érték_ha_hiba) 

Azokban az esetekben érdemes alkalmazni, ha egy nagyobb tömbön belül alkalmazott képlet néhány 

esetben #ÉRTÉK eredményt ad különböző okokból (pl. üres cella/ nullát tartalmazó cella adott 

aritmetikai műveletek mellett nem ad eredményt/ nem felel meg a feltételeknek az adott cella) 

SZÖVEG.TALÁL 
– Adott szöveg megkeresése egy cellán belül. Egy karaktersorozatban egy másikat keres, és 

eredményül az első karakterlánc első karakterének helyét adja a második karakterlánc elejétől 

számítva 

Szintaxis: SZÖVEG.TALÁL(keres_szöveg; szöveg; [kezdet]) 

A függvény különbséget tesz a kis- és a nagybetűk között, és nem teszi lehetővé a helyettesítő 

karakterek használatát. (Ha a keresés közben nem szeretnénk megkülönböztetni a kis- és a 

nagybetűket, vagy ha helyettesítő karaktereket kell használni, akkor a SZÖVEG.KERES függvényt 

kell használni. 

 

SZORZATÖSSZEG 
– Tömbök összetevőinek páronként történő összeszorzása, majd a szorzatértékek összeadása 

Szintaxis: SZORZATÖSSZEG(tömb1; [tömb2]; [tömb3]; ...) 

A SZORZATÖSSZEG függvénnyel azonos méretű tömbök kezelhetők. A kijelölt tömbök 

páronkénti értékeinek kiszámítja a szorzatát, majd ezeket adja össze. 

Példa: A gyártósori műveleti időket a munkahelyi ciklusidőkből és az ott dolgozó operátorok 

számából is ki tudjuk számolni (önellenőrzés). 

Tudjuk, hogy a C gyártósori műveleti idők összege 642 másodperc. Ezt a saját számításaink 

ellenőrzésére is felhasználhatjuk. Nézzük meg milyen eredményt kapunk a munkahelyi 

dolgozószám és a  munkahelyi ciklusidők szorzatösszegéből. 

=SZORZATÖSSZEG(G35:N35;G42:N42) 



  Anyagmozgatási és Logisztikai Rendszerek Tanszék 
 Bevezetés a lean szemléletbe Házi Feladat segédlet 

Budapesti Műszaki és Gazdaságtudományi Egyetem  

Közlekedésmérnöki és Járműmérnöki Kar  6 

 

 

Az eredmény 642 lett, így a számításban feltételezhetően nincs hiba. 

FKERES 
– Egy táblázatban/tömbben szereplő érték függőleges megkeresése, majd a tőle jobbra lévő, adott 

oszlopban megtalálható hozzárendelt érték visszaadása eredményül. 

Szintaxis: =FKERES(A keresendő érték, a keresett hely, a visszaadni kívánt értéket tartalmazó 

tartomány oszlopszáma, közelítő vagy pontos egyezés – 1/IGAZ vagy 0/HAMIS). 

VKERES 
– Értéket keres a táblázat vagy egy értékekből álló tömb legfelső sorában, majd visszaadja a táblázat 

vagy tömb megfelelő értékét a találat oszlopának abból a sorából, amelyet meghatároz. Tehát akkor 

alkalmazhatjuk, ha az összehasonlítandó értékek az adattáblázat tetején sorban helyezkednek el, és 

alattuk meghatározott számú sorban szeretne keresni, 

Szintaxis: =VKERES(keresési_érték; tábla; sor_szám; [tartományban_keres]) 

 

 

 

SZÖVEGBŐL OSZLOPOK 
– Több szöveget tartalmazó cella tartalmának szétosztása több oszlopba 

Adatok→Szövegből oszlopok→Tagolt→Tovább 

Több lehetőségünk is van a szétosztás elvégzésére (-mi a határolóelem). 

Jelen esetben szóköz választja el azokat a cellaelemeket egymástól, amiket külön oszlopokban 

szeretnénk megjeleníteni. 2. lépés: Szóköz kiválasztása→Tovább 

 

A 3. lépésnél ki kell választanunk az oszlopok adattípusát (ez maradhat Általános esetünkben, de a 

cella tartalma határozza meg). Majd meg szükséges adni a Rendeltetési helyet, ahova szeretnénk, 

hogy az Excel megjelenítse a tagolt oszlopokat. (Elegendő az első oszlop helyének megadása) 

A példában a rendeltetési hely közvetlenül a következő oszlop. 

A Befejezés gombra kattintva a következő ábrát láthatjuk eredményül: A Műveletek kódja oszlop 

mellé kerültek a szétosztott oszlopok. 

Kapcsolati mátrix készítése (hahiba, vkeres, ha függvények alkalmazása) 
Kapcsolati mátrix készítése: melyik művelethez, mely alapanyagok tartoznak (ez csak egy példa- 

meghatározási módszer, természetesen másképp is el lehet készíteni Excelben). 

Azután készíthetünk kapcsolati mátrixot, ha elvégeztük az szövegből oszlopok feladatot, és 

megkaptuk külön oszlopokra bontva, hogy az egyes alapanyagok milyen műveletekben fordulnak 

elő. 

Mi a kapcsolati mátrix? A kapcsolati mátrix nevéből eredően a kapcsolatok létezését mutatja meg 

táblázat formájában. Így 0 vagy 1 értéket tartalmaz 1-1 cella. Ebben a példában egy olyan táblázatot 

szükséges készíteni, amelynek a soraiban az egyes alapanyagok, míg az oszlopokban az egyes 

műveletek kapnak helyet. A kapcsolati mátrix segítségével könnyedén meg tudjuk határozni, hogy 

az egyes műveletek során hány alapanyag épül be a termékbe, valamint némi átalakítást követően 

azt is meghatározhatjuk, hogy hány KLT-s és hány GLT-s alapanyag van köztük. 

Kiinduló adatok: Az alapanyagokat és a hozzájuk tartozó műveleteket tartalmazó tömb. 

 

 

Kapcsolati mátrix elkészítési lépései: 

1. lépés: Táblázat első oszlopának feltöltése (sorok): Alapanyagok- A001-A040 

2. lépés: Táblázat első sorának feltöltése (oszlopok): Műveletek- C1…QC* 

3. lépés: Függvény elkészítése 

Meg kell vizsgálnunk soronként/alapanyagonként az F:N oszlopok tartalmát. Ha az adott 

sorban megtaláljuk a keresett műveletet ( Táblázatunk felső sora alapján), akkor a művelet 

jelenjen meg az adott cellában. Mivel az egyes műveletekhez nincs szükségünk az összes 

alapanyagra, ezért ahol nincs kapcsolat, ott egy szóköz jelenjen csak meg (’’ ’’). 

Ez alapján a VKERES függvény segítségével megvizsgáljuk alapanyagonként, hogy az 

adott művelethez hozzátartozik-e vagy sem. Ahol nem tartozik hozzá, ott a függvénynek 

hibára kell futnia, amit HAHIBA függvény segítségével küszöbölünk ki. 

 

=HAHIBA(VKERES(Q$1;$A2:$N2;1;HAMIS);" ") 

 

Ahhoz, hogy a függvényünk az egész mátrixon belül alkalmazható legyen fixálni kell a 

VKERES függvény keresési tartomány oszlopait (csak sorban haladjon lefelé és jobbra ne 

tolódjon el a tartomány). Valamint a VKERES függvény keresési értékén belül a cella 

sorát kell rögzítenünk, hogy a függvény csak a táblázat első sorában lévő műveleteket 

keresse meg. 

Rögzítés: A függvényeken belül $ segítségével tudunk sor / oszlop értékeket vagy egész 

tömböket rögzíteni.  

Pl: A$1: sor rögzítése ; $A1: oszlop rögzítése ; $A$1: cella rögzítése 

 

 

 
 

 

4. lépés: Ha a függvényt végighúzzuk a táblázat egészén, akkor kapunk egy táblázatot, ami 

tartalmazza, hogy az egyes alapanyagok, mely műveleteknél szükségesek. Látható, hogy ez 

még nem a kapcsolati mátrix, mivel nem 0 és 1 értékek szerepelnek benne, hanem a 

műveletek elnevezése. 

5. lépés: Új táblázat készítése (Meglévő táblázat első sorának és oszlopának átmásolása) 

6. lépés: Új függvény készítése, hogy a táblázatban 0 és 1 értékek legyenek a kapcsolat 

létezését mutatva 

 

Az új táblázatban létre kell hozni, egy olyan függvényt, ami azokban az esetekben, ha a 

cellában a felső(első) sorban lévő művelet van, akkor 1-et ad eredményül. Míg a többi 

esetben 0-t. 

 

=HA(BE$1=Q2;1;0) 

 

Ahhoz, hogy a függvény alkalmazható legyen a táblázaton belül, rögzíteni kell a logikai 

vizsgálat egy elemét. Azt szeretnénk, ha az első táblázat összes elemét az új táblázat felső 

(első) sorának értékeivel (műveleteivel) hasonlítaná össze, emiatt az új táblázatban 

szereplő cellának a sorértékét kell rögzíteni (így a függvény a táblázaton belüli 

végighúzással nem fog sort váltani; csak az első sorban fog végigfutni). 

 

 
 

7. lépés: Ha a függvényt végighúzzuk a táblázat egészén, akkor megkapjuk a kapcsolati 

mátrixot. 

Kanban kártyák számának meghatározása 
1. lépés: A kapcsolati mátrix felvétele alapanyagok és munkahelyek között 

2. lépés: A munkahelyenkénti beépülések kiszámítása 1 termék esetében 

3. lépés: A munkahelyenkénti beépülések kiszámítása 1 napra vonatkozóan 

4. lépés: Az óránkénti anyagigény kiszámítása (D) 

5. lépés: Az kanban kártyák számának (N) becsült értékének felvétele 

6. lépés: A becsült N értékhez tartozó kanban köridejének kiszámítása (tc) 

7. lépés: Az N becsült értékekben a módosítások elvégzése * 

8. lépés: Milkrun járat meghatározása (T és β) 

9. lépés: Végleges kanban kártyák számának meghatározása (N*, N**) 

Az alábbi mintapéldában 2 gyártósor munkahelyeinek az alapanyagigény értékei kerülnek 

meghatározásra. A kiinduló adatok között szerepel az alapanyagokat összefoglaló táblázat és a 

műveletek munkahelyi kiosztása. (Természetesen itt is igaz, hogy másképpen is ki lehet számítani 

a feladatot. Ez csak egy minta a házi feladathoz.) 

Négy alapanyag szerepel a példában. A táblázat alapján pedig ismert az alapanyagok típusa 

(KLT/GLT), a tárolóegységben elhelyezett mennyiségi értéke, az egy termékbe beépülő darabszám 

és az, hogy mely műveleteknél kerülnek beépítésre.  

 

Látható, hogy összesen 7 darab E gyártósori művelet és 4 db F gyártósori művelet van. Ezek pedig 

4 db E gyártósori munkahelyre és 2 db F gyártósori munkahelyre lettek kiosztva az alábbi táblázat 

szerint. 

 

 

1. lépés: A kapcsolati mátrix felvétele alapanyagok és munkahelyek között 

A két táblázat ismeretében elkészíthető a kapcsolati mátrix az előző fejezetben leírtak alapján 

(Kapcsolati mátrix készítése (hahiba, vkeres, ha függvények alkalmazása)).  

2. lépés: A munkahelyenkénti beépülések kiszámítása 1 termék esetében 

A kapcsolati mátrix és a munkahelykiosztás segítségével meghatározható a munkahelyeken történő 

előfordulása az egyes alapanyagoknak. Tehát megkaphatjuk, hogy hány műveletbe épül be a 

vizsgált alapanyag az adott munkahelyen.  

Ahhoz, hogy ezt ki tudjuk számolni, meg kell vizsgálni, hogy az adott munkahely milyen 

műveletekből épül fel. Majd meg kell vizsgálni, hogy a munkahelyi műveleteknél beépítésre kerül-

e az adott alapanyag. A képletben felhasznált függvények:  

Vkeres: Az adott munkahelyhez tartozó műveleteket megkeresi a kapcsolati mátrixon belül (1-et ad 

vissza, ha az adott alapanyag az adott munkahelyen előfordul a megkeresett műveletek alapján, és 

0-t ,ha az adott alapanyaghoz nem tartozik a vizsgált művelet) 

A munkahelykiosztást tartalmazó táblázatban az látható, hogy az egyes munkahelyeken belül 1..3 

közötti művelet kaphat helyet. Tehát egy munkahely esetében maximum 3x kell megkeresni a 

kapcsolati mátrixban az adott munkahelyi műveleteket. Emiatt 3x kell vkeres függvényt alkalmazni 

és mivel a munkahelyeken történő előfordulást kell megadnunk, ezért összegeznünk kell a 

visszaadott 0 vagy 1 értékeket.  

Hahiba: A hahiba függvény alkalmazására azért van szükségünk, mert előfordul olyan munkahely 

is, ahol 3-nál kevesebb művelet kapott helyet, és a végleges képlet így nem fog hibára futni, ha 

végighúzzuk a táblázaton.  

=VKERES(E$10;$M$9:$X$13;$C17;HAMIS)+HAHIBA(VKERES(E$11;$M$9:$X$13;$C17;HA

MIS);0)+HAHIBA(VKERES(E$12;$M$9:$X$13;$C17;HAMIS);0) 

Segédoszlop: Az 1,2,3,4,5 értékeket tartalmazó segédoszlopok abban segítenek, hogy a vkeres 

függvény a megfelelő alapanyaghoz tartozó sorban található 0 vagy 1 értéket adja vissza a kapcsolati 

mátrixból.  

A képlet táblázaton belüli alkalmazásához rögzíteni kell a kapcsolati mátrixot tartalmazó tömb 

egészét, a segédoszlop értékeinél az oszlopot, a munkahelyi kiosztás táblázatán belül a vizsgált 

műveleteknél pedig a sort. 

 

Az eredményt tartalmazó táblázatból az olvasható le, hogy a két gyártósor munkahelyein hány 

műveletnél kerül felhasználása az adott alapanyag.  

A következő lépés az egy termékbe való beépülések megadása, így megtudhatjuk, hogy egy termék 

elkészítésekor mekkora mennyiségű alapanyag fogy az egyes munkahelyeken. Ezt elvégezhetjük 

egyszerű szorzással, ha rögzítjük az Egy termékbe beépülő darabszám az adott műveletben 

értékeinek az oszlopértékét, vagy természetesen egy fkeres függvényt is írhatunk a megfelelő 

alapanyagokra vonatkozóan. (Az előbbi lépés kihagyásával is megkapható ez a táblázat. Ilyenkor a 

vkeres által visszaadott 0 vagy 1 értékeket össze kell szorozni egy fkeres függvény segítségével a 

beépülő mennyiségekkel.) 

 

 

A kapott eredményeket a következő táblázat tartalmazza.  

 
 

3. lépés: A munkahelyenkénti beépülések kiszámítása 1 napra vonatkozóan 

Ahhoz, hogy ki tudjuk számolni az egy napra vonatkozó beépüléseket munkahelyenként, először 

meg kell adni gyártósoronként a termelt mennyiségeket (kapacitásokat). Ezek az értékek minden 

esetben nagyobbak vagy egyenlőek a napi kiszolgálandó igénnyel. A példában a napi igény és 

a sori kapacitások a következőképp alakultak.  

Produktív időalap [h] 15 

Napi igény [db] 600 

Kapacitás E soron 606,56 

Kapacitás F soron 620,93 

A napi beépülő anyagok kiszámításához fel kell használni az egy termékbe való beépülések 

táblázatát és a napi számított, lefelé kerekített kapacitásértékeket a megfelelő sorokra vonatkozóan.  

A képletben az adott munkahelyen felhasznált, 1 termékbe beépülő anyagmennyiség értékét meg 

kell szorozni az adott munkahelyre vonatkozó, lefelé kerekített sorkapacitás értékével. A megfelelő 

szorzások elvégzésével a következő táblázatot kapjuk eredményül.  

 
 

4. lépés: Az óránkénti anyagigény kiszámítása (D) 

Az egy nap alatt felhasznált alapanyagok mennyiségi értékének kiszámítása után meghatározható 

az óránkénti fogyás értéke, ami a kanban-szám számítási képletében a D-vel jelölt érték lesz 

esetünkben. Az óránkénti anyagigényt a napi fogyás és a produktív időalap hányadosa fogja 

megadni. Ezt felfelé kell kerekíteni, hogy egész számú fogyást kapjunk (és a biztonság irányába 

mozduljunk el).  

 
 

5. lépés: Az kanban kártyák számának (N) becsült értékének felvétele 

A kanban kártyák számának meghatározása előtt érdemes megvizsgálni, hogy mekkora az egy nap 

alatt felhasznált KLT és GLT tárolók számossága. Ez alapján lehetséges annak az eldöntése, hogy 

az adott alapanyag az adott munkahelyen milyen módon kerül feltöltésre. A házi feladat feladatlapja 

alapján az üzem vezetősége a napi egyszeri feltöltést támogatja a raktárból. Így azokat az 

alapanyagokat, amelyekből egy nap alatt 6-nál több KLT, vagy 2-nél több GLT fogy, 

szupermarketben kell tárolni és onnan kell biztosítani a napi többszöri feltöltést. 

A napi felhasznált KLT/GLT számok meghatározásához a napi felhasznált alapanyagokat 

tartalmazó táblázatot és a kiinduló táblázatot kell felhasználni, amelyből kiderül, hogy egy tárolóban 

hány darab van az adott alapanyagból. 

A napi felhasznált KLT/GLT darabszám = a napi felhasznált anyagmennyiség és az adott tárolóban 

tárolt alapanyag darabszámának hányadosa. Majd a kapott értéket felfelé kerekítjük.  

Az összes felhasznált KLT/GLT mennyiség meghatározását követően eldönthető az alapanyag 

tárolási helye. Itt határozzuk meg, hogy az adott termék tárolható-e a munkahelyen, vagy 

szupermarket kialakítására is szükség van. A házi feladatban egy munkahelyen maximum 6 KLT és 

maximum 2 GLT tárolható egy termékből. A tárolás helyének eldöntésére egy példát láthattok a 

dokumentumban, azonban ezt többféleképpen is meg lehet határozni (szűrés, egyéb számítás (ha, 

és, vagy függvényekkel), feltételes formázás...). 

Az első lépés a termékek munkahelyi előfordulásainak az összegzése. Tehát termékenként 

megszámoljuk, hány munkahelyen szerepel az adott alapanyag 0-tól nagyobb értékkel.  

=DARABHATÖBB(F45:K45;">0") 

A következő oszlopban azt vizsgáljuk meg, hogy hány munkahelyen valósítható meg az adott 

alapanyag munkahelyi tárolása. A függvénnyel először megnézzük egy ha függvény segítségével, 

hogy a termék KLT-ben tárolható-e. Ha igen, akkor összeszámolja azokat a munkahelyeket, ahol 

nagyobb mint 0, de 6-nál nem nagyobb KLT szám szerepel. Ha nem KLT-s, a termék (tehát GLT-

s), akkor pedig összeszámolja a nullánál nagyobb, de 2-nél nem nagyobb munkahelyek 

számosságát.  

=HA(E45="KLT";DARABHATÖBB(F45:K45;">0";F45:K45;"<6");(DARABHATÖBB(F45:K45;

">0";F45:K45;"<2"))) 

A két oszlop eredményeit már csupán össze szükséges hasonlítani. Akkor tárolható munkahelyen 

az adott alapanyag, ha az összes munkahelyen, amelyen az alapanyag előfordul, maximum 6 KLT 

kerül egy nap alatt felhasználásra. Tehát meg kell vizsgálni, hogy a munkahelyi előfordulási szám 

és a munkahelyi tárolási szám egyezik-e. Ha igen, akkor lehet munkahelyi tárolású a termék, ha 

nem, akkor pedig szupermarketben szükséges tárolni. (Előfordulhat olyan eset is, amikor pl. 4 

munkahelyen fordul elő a termék és ebből a 4 munkahelyből 1, 2 vagy 3 munkahelyen is munkahelyi 

tárolású lehetne az alapanyag, mivel 1..6 számú KLT fogy egy nap alatt. Az ilyen esetekben felülírja 

a döntést, hogy 1 munkahely miatt mindenképp szupermarketben is kellene tárolni az adott 

alapanyagot, emiatt úgy döntünk, hogy az összes munkahelyet a szupermarketből fogjuk utántölteni.  

A táblázat alapján látható, hogy az A002 kivételével az összes alapanyagot szupermarketben 

szükséges tárolni.  

A következő feladat a kanban kártyák számának a kiindulási értékeinek a felvétele. Erre is több 

lehetőségünk van. Mivel ezek kiinduló értékek lesznek, csak azt befolyásolja a kiválasztásuk, hogy 

milyen mértékben kell a későbbiekben módosítani őket.  

Az első lehetőségünk, ha minden alapanyagnál -azokon a munkahelyeken, ahol beépítésre kerül- 

felveszünk egy 6-os értékű kanban számot (GLT-s alapanyagoknál 2 értéket). 

A második lehetőség, valamilyen arány definiálása, ami súlyozni fogja a kanban értékeit. Azt 

azonban fontos figyelembe venni, hogy esetünkben ez az érték 2 és 6 között ingadozhat csak, 

mivel 6-nál több KLT-s termék nem tartható munkahelyen. GLT esetében pedig csak 2 értéket 

vehet fel. Az arány meghatározásának egyik módja az óránkénti alapanyagfelhasználás segítségével 

történik. Először kiválasztjuk a legnagyobb D értéket a KLT-s termékek közül. Ez egy egyszerű 

max függvény segítségével megadható. (A példában ez a 243-as értéket fogja visszaadni, ami az 

A002-höz tartozik.) Ezt elosztva a kanban kártyák számának maximális értékével (6) megkapjuk az 

arányt, ami ebben a példában 40,5 lesz. 

A kalkulált arány segítségével egyszerűen kioszthatjuk a kiinduló értékeket a KLT-s 

alapanyagokhoz. Ha az óránkénti anyagigény 0, akkor 0 értéket kell visszaadni. Ha nem nulla az 

óránkénti anyagigény, vagyis beépülésre kerül az alapanyag az adott munkahelyen, akkor az 

óránkénti igényt elosztjuk az aránnyal. A max függvény segítségével pedig azt küszöböljük ki, hogy 

az N értéke 1 legyen.  

=HA(F54=0;0;MAX(KEREK.FEL(F54/$B$60;0);2)) 

 

GLT-s termékeknél pedig egy egyszerűbb ha függvényt alkalmazhatunk. (Ha az óránkénti 

anyagigény =0, akkor 0 értéket adjon vissza, egyébként pedig 2-t.) 

6. lépés: A becsült N értékhez tartozó kanban köridejének kiszámítása (tc) 

Vegyük elő a kanban kártyák számának kiszámításához alkalmazható képletet és azonosítsuk be, 

milyen adatok állnak jelenleg a rendelkezésünkre (1). Az N értékeket kiszámítottuk az 5. lépésben, 

a D értékeket pedig a 4. lépésben. A K, 𝜹, 𝝁 értékek pedig adottak. Így a képlet átrendezésével 

kiszámíthatjuk a köridőket (tc)a kiinduló kanban értékek mellett.  

(1) 𝑁 =
𝐷∗(1+𝛿)∗𝑡𝑐∗(1+𝜇)

𝐾
 

(2) 𝑡𝑐 =
𝑁∗𝐾

𝐷∗(1+𝛿)∗(1+𝜇)
 

A képlet átrendezésével (2) az N és a K kerül a számlálóba, míg a D, 𝜹, 𝝁 a nevezőbe kerül. Az 

óránkénti anyagfelhasználási táblázatban vannak 0 értékek is, emiatt hahiba függvényt is 

alkalmazni szükséges, hogy ne adjon vissza ZÉRÓOSZTÓ hibát az MS Excel. A megfelelő 

alapanyaghoz tartozó K értékeket fkeres segítségével adhatjuk meg, ami a legelső táblázatban keresi 

meg az alapanyaghoz tartozó DB/Tárolóegység értékeket (vagy egyszerűen az alapanyagok 

cikkszáma mellé felvesszük új oszlopként a hozzájuk tartozó K értékeket      . Fontos, hogy mindkét 

esetben rögzíteni kell vagy a tömböt, vagy az oszlopot). A nevezőben pedig a rögzített 𝜹 és 𝝁 értékek 

mellett ki kell jelölni az adott alapanyag, adott munkahelyéhez tartozó óránkénti anyagigény 

mennyiségét (D). 

=HAHIBA((F64*FKERES($D74;$D$2:$F$5;3;HAMIS))/(F54*(1+$B$54)*(1+$B$53));0) 

 

 

Most pedig vissza kell ellenőrizni a kapott értékeket. Érdemes a legkisebb értékekre koncentrálni 

először, hogy ez mit is jelent a feladat szempontjából számunkra. Az üzem 8 órás műszakokban 

dolgozik, így a T értéke {0,5 ; 1; 2; 4; 8} lehet. A β pedig 2 és 3 értékeket vehet fel. Tehát a tc 

minimális értéke 1 órának adódik. Azt látjuk, hogy az A001 alapanyaghoz tartozó kanban köridő 

értékek kisebbek, mint 1, emiatt változtatni kell a kanban kártyák számán. Ezt fogjuk elvégezni a 

következő lépésben.   

(Ha nem az arányosított kanban számokkal számolunk kiindulásként, hanem a lehetséges 

legnagyobb értékekkel (KLT: 6, GLT:2), akkor a következő táblázatot kapjuk a köridő értékekre. 

Itt is a legkisebb köridőt kell megvizsgálnunk. Ez az érték 1,88 órának adódik, emiatt a T és β 

értékeket úgy kell megválasztani, hogy 2 óra alatti számot adjon a szorzatuk ( marad a tc minimális 

értéke 1 óra, mint a másik kiinduló N értékekkel való számításnál). 

7. lépés: Az N becsült értékekben a módosítások elvégzése * 

Az N értékek módosítását érdemes egy új munkalapon elvégezni (munkalap másolata), így 

megmaradnak a kiinduló N értékeink és az új munkalapon automatikusan kiszámításra kerülnek a 

tc értékek is. 

Az A001 alapanyagnál elkezdjük növelni a KLT számot, és akkor állunk meg, ha 1 feletti értéket 

kapunk. Így a módosított KLT-s kanban kártyaszámokat az első alapanyagnál 4-re kell növelni. A 

számok átírásával a következő táblázatokat kapjuk eredményül. 

 

 

Itt már csak 1 feletti értékeket kapunk köridőnek.  

 

8. lépés: Milkrun járat meghatározása (T és β) 

 

Mivel a köridő 1 órának adódott, emiatt a következő lépésben leszűkülnek a lehetőségeink a T és β 

megválasztására. T=0,5 és β=2 értékekkel kapunk csak 1 órás köridőt. Sikerült meghatározni a 

köridőt és a milkrun járatot. Mivel a β 2-es értéket kapott (és feltehetően önkiszolgáló a járat), emiatt 

növelni kell a járat bizonytalansági tényezőjét.  

 

9. lépés: Végleges kanban kártyák számának meghatározása (N*, N**) 

 

A köridő és a milkrun járat tulajdonságainak meghatározása után, ebben a lépésben vissza kell 

ellenőrizni a kanban kártyák számát. Újra felhasználásra kerül az (1) képlet, most átalakítás nélkül. 

Ismerjük a tc értékét (8. lépés) és a D értékeket (4. lépés). A K, 𝜹 értékek adottak voltak és nem 

történt módosítás. A 𝝁 értékét pedig 0,5-re növeltük. Így kiszámolhatóak az N* értékek.  

A végleges N* értékeknél fontos, hogy 1 értéket nem kaphatunk eredményül, így vagy egy lépésben 

egy olyan függvényt kell megírni, ami kezeli ezt. Vagy több lépésben pl.  át kell alakítanunk az N* 

értékeket, hogy 0 esetén 0 maradjon, de 1 esetén 2-re váltson.  

1 lépéses függvény felépítésére egy példa: Először kiszámoljuk a képlet segítségével az N* értékét 

(minden adott hozzá), ezt követően ezt felkerekítjük és megvizsgáljuk, hogy kisebb-e, mint 2. Ha 

kisebb, akkor a 2-es értéket kérjük vissza ( max függvény). Ezt követően, mivel nem szeretnénk, 

hogy azokon a munkahelyeken is legyen olyan alapanyagból 2 KLT vagy GLT, ahol nem is kerül 

beépítésre, emiatt megnézzük az óránkénti beépülések számát. Ha 0 értéket találunk, akkor 0-t adjon 

vissza a függvény, egyébként pedig a kiszámított, kerekített és maximalizált N* értékeket.  

=HA(D=0;0;MAX(KEREK.FEL(((D*(1+ μ)*(1+ δ)* T *β)/K);0);2)) 

 


